1 Calcul approché d'intégrales

Le but de cette partie est le calcul approché de l'intégrale d'une fonction f sur l'intervalle $[a,b]$ qu'on divise en n intervalles de même longueur.
On supposera d'abord que sur chacun de ces petits intervalles la fonction est constante et ensuite, que sur chacun de ces petits intervalles la fonction est affine, c'est à dire que son graphe est une droite.
On essaiera avec la fonction « `sinus` », qu'on appelera F, entre 0 et $\frac{\pi}{2}$.
Ce qui donne :

```
> F:=x->sin(x);
```

$$F := \sin$$

1.1 Intégration en « batons »

On va faire cette intégration de deux façons, en supposant d'abord la fonction constante ayant la valeur de f à la borne droite de l'intervalle, puis la fonction constante ayant la valeur de f à la borne gauche de l'intervalle.
On aura à chaque fois 4 paramètres : f,a,b,n.
On utilisera une variable locale S dans laquelle on ajoutera à chaque fois l'aire du petit rectangle considéré, et k qui indique sur quel intervalle on travaille. S sera ainsi initialisée à 0, et tous les calculs se feront en valeur approchée.
Ce qui donne en sommant sur la valeur à droite de l'intervalle :

```
> Integ1:=proc(f,a,b,n) local k,S; S:=0; for k from 1 to n do
S:=S+evalf(f(a+k*(b-a)/n)*(b-a)/n) od; S end;
```

Integ1 := **proc**(f, a, b, n)
local k, S;
$S := 0$; **for** k **to** n **do** $S := S + \text{evalf}(f(a + k * (b - a)/n) * (b - a)/n)$ **end do** ; S
end proc

```
> Integ1(F,0,Pi/2,50);
```

$$1.015625716$$

Et en sommant sur la valeur à gauche de l'intervalle :

```
> Integ2:=proc(f,a,b,n) local k,S; S:=0; for k from 0 to n-1 do
S:=S+evalf(f(a+k*(b-a)/n)*(b-a)/n) od; S end;
```

Integ2 := **proc**(f, a, b, n)
local k, S;
$S := 0$;
for k **from** 0 **to** $n - 1$ **do** $S := S + \text{evalf}(f(a + k * (b - a)/n) * (b - a)/n)$ **end do** ;
S
end proc

```
> Integ2(F,0,Pi/2,50);
```

$$.9842097891$$

1.2 Sommation en «trapèzes »

Si on fait une petite figure, on s'aperçoit que la sommation en trapèze est la moyenne des deux précédentes !
Ce qui donne, en vous laissant analyser le problème :

```
> Integ3:=proc(f,a,b,n) local k,S; S:=evalf((f(a)+f(b))/2*(b-a)/n);
for k from 1 to n-1 do S:=S+evalf(f(a+k*(b-a)/n)*(b-a)/n) od; S end;
```

Integ3 := **proc**(f, a, b, n)
local k, S;
$S := \text{evalf}(1/2 * ((f(a) + f(b)) * (b - a))/n)$;
for k **to** $n - 1$ **do** $S := S + \text{evalf}(f(a + k * (b - a)/n) * (b - a)/n)$ **end do**;
S
end proc

```
> Integ3(F,0,Pi/2,50);
```

.9999177523

2 Polynômes de Tchebytcheff

Le but de cette partie est de calculer les polynômes de Tchebytcheff. Observons d'abord :

- $\cos t = \cos t$
- $\cos 2t = 2\cos^2 t - 1$
- $\cos 3t = 4\cos^3 t - 3\cos t$

On peut penser alors et on montre facilement par récurrence que $\cos nt$ est un polynôme en $\cos t$.
Ces polynômes vérifient la relation de récurrence :

$$T_n(x) = 2x\,T_{n-1}(x) - T_{n-2}(x)$$

car $\cos nt + \cos(n-2)t = 2\cos t\cos(n-1)t$.
On ajoute les conditions initiales et la suite est déterminée : $T_0(x) = 1$ et $T_1(x) = x$.
Il ne reste qu'à écrire la procédure à 1 paramètre n.

```
> Tcheb:=proc(n) local p,q,r; p:=x;q:=2*x**2-1; from 3 to n do
r:=collect(-p+2*x*q,x); p:=q;q:=r od; r end;
```

Tcheb := **proc**(n)
local p, q, r;
$p := x$;
$q := 2 * x^2 - 1$;
from 3 **to** n **do** $r := \text{collect}(-p + 2 * x * q, x)$; $p := q$; $q := r$ **end do**;
r
end proc

On va suivre, sur le tableau ci-dessous, l'exécution de se programme en regardant les valeurs des différentes variables à **l'entrée** et au **milieu** de la boucle, c'est à dire juste après le `do` et entre le `collect(-p+2*x*q,x)` et le `p := q; q := r`.
Pour la dernière boucle, on donne aussi les valeurs à la **sortie** de la boucle, qui ne sont que les valeurs à l'entrée de l'itération suivante.

	Valeur des variables dans la boucle :						
Variable Maple	Entrée	Milieu	Entrée	$\cdots$	Entrée	Milieu	Sortie
p	$T_1(x)$	$T_1(x)$	$T_2(x)$		$T_{n-2}(x)$	$T_{n-2}(x)$	$T_{n-1}(x)$
q	$T_2(x)$	$T_2(x)$	$T_3(x)$		$T_{n-1}(x)$	$T_{n-1}(x)$	$T_n(x)$
r		$T_3(x)$	$T_3(x)$		$T_{n-1}(x)$	$T_n(x)$	$T_n(x)$

Essayons cette procédure pour deux valeurs de n :

```
> Tcheb(3);
```

$$4\,x^3 - 3\,x$$

```
> Tcheb(11);
```

$$1024\,x^{11} - 2816\,x^9 + 2816\,x^7 - 1232\,x^5 + 220\,x^3 - 11\,x$$